

Diferencialne operacije

```
In[1]:= f[t_] := a t^2 + b t + c
```

```
In[2]:= res = Solve[{f[-h] == f1, f[0] == f2, f[h] == f3}, {a, b, c}][[1]]
```

```
Out[2]= {a -> - $\frac{-f_1 + 2 f_2 - f_3}{2 h^2}$ , b -> - $\frac{f_1 - f_3}{2 h}$ , c -> f2}
```

```
In[3]:= (* prvi odvod - enakomeren korak *)
{HoldForm[f'[-h]] == Simplify[f'[-h] /. res],
 HoldForm[f'[0]] == Simplify[f'[0] /. res],
 HoldForm[f'[h]] == Simplify[f'[h] /. res]} // TableForm
```

Out[3]//TableForm=

```
f'[-h] == - $\frac{3 f_1 - 4 f_2 + f_3}{2 h}$ 
f'[0] ==  $\frac{-f_1 + f_3}{2 h}$ 
f'[h] ==  $\frac{f_1 - 4 f_2 + 3 f_3}{2 h}$ 
```

```
In[4]:= (* drugi odvod - enakomeren korak *)
{HoldForm[f''[-h]] == Simplify[f''[-h] /. res],
 HoldForm[f''[0]] == Simplify[f''[0] /. res],
 HoldForm[f''[h]] == Simplify[f''[h] /. res]} // TableForm
```

Out[4]//TableForm=

```
f''[-h] ==  $\frac{f_1 - 2 f_2 + f_3}{h^2}$ 
f''[0] ==  $\frac{f_1 - 2 f_2 + f_3}{h^2}$ 
f''[h] ==  $\frac{f_1 - 2 f_2 + f_3}{h^2}$ 
```

```
In[5]:=
```

```
In[6]:= s = Import["~/vaje/rovf12/vaja5/kamen.dat", {"Data", All, {1, 2}}];
v = Import["~/vaje/rovf12/vaja5/kamen.dat", {"Data", All, {1, 3}}];
a = Import["~/vaje/rovf12/vaja5/kamen.dat", {"Data", All, {1, 4}}];
```

```

In[9]:= odvod1[s_] := Module[{n, h},
  n = Length[s];
  h = s[[2, 1]] - s[[1, 1]];
  Join[{{s[[1, 1]], -(s[[3, 2]] - 4 s[[2, 2]] + 3 s[[1, 2]]) / (2 h)}},
    Table[{s[[i, 1]], (s[[i + 1, 2]] - s[[i - 1, 2]]) / (2 h)}, {i, 2, n - 1}],
    {{s[[n, 1]], (s[[n - 2, 2]] - 4 s[[n - 1, 2]] + 3 s[[n, 2]]) / (2 h)}}]
]

odvod2[s_] := Module[{n, h},
  n = Length[s];
  h = s[[2, 1]] - s[[1, 1]];
  Join[{{s[[1, 1]], (s[[3, 2]] - 2 s[[2, 2]] + s[[1, 2]]) / h^2}},
    Table[{s[[i, 1]], (s[[i + 1, 2]] - 2 s[[i, 2]] + s[[i - 1, 2]]) / h^2}, {i, 2, n - 1}],
    {{s[[n, 1]], (s[[n, 2]] - 2 s[[n - 1, 2]] + s[[n - 2, 2]]) / h^2}}]
]


integral[s_] := Module[{n, h, int},
  n = Length[s];
  h = s[[2, 1]] - s[[1, 1]];
  int = {{s[[1, 1]], 0}};
  For[i = 2, i ≤ n, ++i,
    int = Append[int, {s[[i, 1]], int[[-1, 2]] + 0.5 h (s[[i - 1, 2]] + s[[i, 2]])}]
  ];
  int
]

In[12]:= SetOptions[ListPlot, Joined → True, ImageSize → 5 × 72,
  PlotStyle → Directive[Thickness[Medium]],
  TicksStyle → Thickness[Medium], AxesStyle → Thickness[Medium],
  LabelStyle → Directive[FontFamily → "Helvetica", FontSize → 12]];


In[13]:= Needs["PlotLegends`"]
SetOptions[Legend, LegendShadow → False,
  LegendBorder → False, LegendSpacing → 0, LegendSize → {0.6, 0.2}];
st[x_] := Style[x, FontFamily → "Helvetica", FontSize → 12]

In[16]:= ListPlot[{v, odvod1[s]},
  AxesLabel → {"Čas (s)", "Hitrost (m/s)"}, PlotMarkers → {Automatic, 8},
  PlotLegend → {st["Točno"], st["Izračunano"]}, LegendPosition → {0, -0.4}]


```



```
In[17]:= ListPlot[{a, odvod2[s]},
  AxesLabel → {"Čas (s)", "Pospešek (m/s2)"}, PlotMarkers → {Automatic, 8},
  PlotLegend → {st["Točno"], st["Izračunano"]}, LegendPosition → {0.3, 0.3}]
```




```
In[18]:= ListPlot[{v, integral[a]},
  AxesLabel → {"Čas (s)", "Hitrost (m/s)"}, PlotMarkers → {Automatic, 8},
  PlotLegend → {st["Točno"], st["Izračunano"]}, LegendPosition → {0, -0.4}]
```



```

In[19]:= ListPlot[{s, integral[integral[a]]},
  AxesLabel -> {"Čas (s)", "Pot (m)"}, PlotMarkers -> {Automatic, 8},
  PlotLegend -> {st["Točno"], st["Izračunano"]}, LegendPosition -> {-0.7, 0}]

```



Naloga 1: Nariši graf diferencialne upornosti dU/dI za tokovno odvisnost v datoteki *Korozija.dat*. To je meritev karakteristike I - U za kovinsko elektrodo v določeni korozivni raztopini (prvi stolpec U [mV], drugi I [A]).

Naloga 2: Datoteka *body_acc.txt* vsebuje podatke o kartezičnih komponentah vektorja pospeška premikajoče se osebe (komponente x , y in z pospeška v enotah $g=9.81 \text{ m/s}^2$ so v prvem, drugem in tretjem stolpcu datoteke). Akcelrometer v mobilnem telefonu, ki ga je oseba nasila na pasu, je podatke o pospešku zapisoval s frekvenco 50 Hz. Nariši graf položaja osebe v odvisnosti od časa.

Naloga 3: V datoteki *prevajanje.dat* je podan temperaturni profil v 0.1 m debeli steni v odvisnosti od časa. V prvem stolpcu je podana oddaljenost (v metrih) merilne točke od levega roba stene, v ostalih stolpcih pa so temperature (v stopinjah Celzija) na teh mestih ob časih od 0 s do 180 s s korakom 1 s. (a) Kako se s časom spreminjata toplotna tokova na obeh robovih stene? (b) Ob času $t=1 \text{ min}$ primerjaj krajevno odvisnost odvoda temperature po času s krajevno odvisnostjo drugega odvoda temperature po kraju.